
F07 – Linear Equations (LAPACK)

F07BDF (SGBTRF/DGBTRF) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F07BDF (SGBTRF/DGBTRF) computes the LU factorization of a real m by n band matrix.

2 Specification

SUBROUTINE F07BDF(M, N, KL, KU, AB, LDAB, IPIV, INFO)
ENTRY sgbtrf(M, N, KL, KU, AB, LDAB, IPIV, INFO)
INTEGER M, N, KL, KU, LDAB, IPIV(∗), INFO
real AB(LDAB,∗)

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine forms the LU factorization of a real m by n band matrix A using partial pivoting, with
row interchanges. Usually m = n, and then, if A has kl non-zero sub-diagonals and ku non-zero super-
diagonals, the factorization has the form A = PLU where:

P is a permutation matrix;
L is a lower triangular matrix with unit diagonal elements and at most kl non-zero elements

in each column; and
U is an upper triangular band matrix with kl + ku super-diagonals.

Note that L is not a band matrix, but the non-zero elements of L can be stored in the same space as the
sub-diagonal elements of A. U is a band matrix but with kl additional super-diagonals compared with
A. These additional super-diagonals are created by the row interchanges.

4 References

[1] Golub G H and van Loan C F (1996) Matrix Computations Johns Hopkins University Press (3rd
Edition), Baltimore

5 Parameters

1: M — INTEGER Input

On entry: m, the number of rows of the matrix A.

Constraint: M ≥ 0.

2: N — INTEGER Input

On entry: n, the number of columns of the matrix A.

Constraint: N ≥ 0.

3: KL — INTEGER Input

On entry: kl, the number of sub-diagonals within the band of A.

Constraint: KL ≥ 0.

4: KU — INTEGER Input

On entry: ku, the number of super-diagonals within the band of A.

Constraint: KU ≥ 0.

5: AB(LDAB,∗) — real array Input/Output

Note: the second dimension of the array AB must be at least max(1,N).

On entry: the m by n band matrix A, stored in rows (kl + 1) to (2kl + ku + 1); the first kl rows
need not be set. More precisely, element ai,j must be stored in AB(kl + ku + i − j + 1, j) for
max(j − ku, 1) ≤ i ≤ min(j + kl, m).

On exit: A is overwritten by details of the factorization: the upper triangular band matrix U with
kl + ku super-diagonals is stored in rows 1 to (kl + ku + 1) of the array, and the multipliers used to
form the matrix L are stored in rows (kl + ku + 2) to (2kl + ku + 1).

6: LDAB — INTEGER Input

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BDF
(SGBTRF/DGBTRF) is called.

Constraint: LDAB ≥ 2× KL + KU + 1.

7: IPIV(∗) — INTEGER array Output

Note: the dimension of the array IPIV must be at least max(1,min(M,N)).

On exit: the pivot indices. Row i of the matrix A was interchanged with row IPIV(i), for
i = 1, 2, . . . ,min(m, n).

8: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

INFO > 0

If INFO = i, ui,i is exactly zero. The factorization has been completed but the factor U is exactly
singular, and division by zero will occur if it is subsequently used to solve a system of linear
equations.

7 Accuracy

The computed factors L and U are the exact factors of a perturbed matrix A + E, where

|E| ≤ c(k)εP |L||U |,

c(k) is a modest linear function of k = kl + ku + 1, and ε is the machine precision. This assumes
k � min(m, n).

8 Further Comments

The total number of floating-point operations varies between approximately 2nkl(ku + 1) and 2nkl(kl +
ku + 1), depending on the interchanges, assuming m = n � kl and n � ku.

A call to this routine may be followed by calls to the routines:

F07BEF (SGBTRS/DGBTRS) to solve AX = B or AT X = B;

F07BGF (SGBCON/DGBCON) to estimate the condition number of A.

The complex analogue of this routine is F07BRF (CGBTRF/ZGBTRF).

9 Example

To compute the LU factorization of the matrix A, where

A =



−0.23 2.54 −3.66 0.00
−6.98 2.46 −2.73 −2.13
0.00 2.56 2.46 4.07
0.00 0.00 −4.78 −3.82


 .

Here A is treated as a band matrix with 1 sub-diagonal and 2 super-diagonals.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F07BDF Example Program Text
* Mark 15 Release. NAG Copyright 1991.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER MMAX, NMAX, KLMAX, KUMAX, LDAB
PARAMETER (MMAX=8,NMAX=8,KLMAX=8,KUMAX=8,

+ LDAB=2*KLMAX+KUMAX+1)
* .. Local Scalars ..

INTEGER I, IFAIL, INFO, J, K, KL, KU, M, N
* .. Local Arrays ..

real AB(LDAB,NMAX)
INTEGER IPIV(NMAX)

* .. External Subroutines ..
EXTERNAL sgbtrf, X04CEF

* .. Intrinsic Functions ..
INTRINSIC MAX, MIN

* .. Executable Statements ..
WRITE (NOUT,*) ’F07BDF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) M, N, KL, KU
IF (M.LE.MMAX .AND. N.LE.NMAX .AND. KL.LE.KLMAX .AND. KU.LE.KUMAX)

+ THEN
*
* Read A from data file
*

K = KL + KU + 1
READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,M)

*
* Factorize A
*

CALL sgbtrf(M,N,KL,KU,AB,LDAB,IPIV,INFO)
*
* Print details of factorization
*

WRITE (NOUT,*)
IFAIL = 0

*
CALL X04CEF(M,N,KL,KL+KU,AB,LDAB,’Details of factorization’,

+ IFAIL)
*

* Print pivot indices
*

WRITE (NOUT,*)
WRITE (NOUT,*) ’IPIV’
WRITE (NOUT,99999) (IPIV(I),I=1,MIN(M,N))

*
IF (INFO.NE.0) WRITE (NOUT,*) ’The factor U is singular’

*
END IF
STOP

*
99999 FORMAT ((3X,7I11))

END

9.2 Program Data

F07BDF Example Program Data
4 4 1 2 :Values of M, N, KL and KU

-0.23 2.54 -3.66
-6.98 2.46 -2.73 -2.13

2.56 2.46 4.07
-4.78 -3.82 :End of matrix A

9.3 Program Results

F07BDF Example Program Results

Details of factorization
1 2 3 4

1 -6.9800 2.4600 -2.7300 -2.1300
2 0.0330 2.5600 2.4600 4.0700
3 0.9605 -5.9329 -3.8391
4 0.8057 -0.7269

IPIV
2 3 3 4

